
In This Session ... 

 This session, which is intended for application 
programmers, covers new and changed key 
system APIs that were introduced to the i 
operating system in V5R4 through 7.1 

 

– 7.1: @ 50 new,  @ 80 changed 

– V6R1: @170 new, @ 250 changed 

– V5R4: @ 90 new, @ 170 changed 

 

 Examples shown will be using free form RPG 

 



4/24/2012 

2 

Copyright Bruce Vining Services LLC, 2010, 2012 

All  Rights Reserved 

What We’ll Cover … 

 Major New V5R4 APIs 

 Major New V6R1 APIs 

 Major New 7.1 APIs 

 Overview of Additional New and Changed 

 Wrap-up 

Message Watching 

 Watch technology provides the ability to be notified in real 

time when a problem occurs 

 Identify what the problem is 

– The symptom of a potential problem 

– ‘Problem’ can be 

• A message and where it is logged  

• A LIC Log 

 Dynamically define (add or remove) the watch points 

 When the problem occurs get notified via an exit point 

 



4/24/2012 

3 

Copyright Bruce Vining Services LLC, 2010, 2012 

All  Rights Reserved 

Message Watching 

 STRWCH, ENDWCH, WRKWCH command 

 QSCSWCH, QSCEWCH APIs 

 Watch for messages (V5R4) on: 

– QSYSOPR 

– QHST 

– *MSGQ 

– Job 
• Specific job 

• Generic job (VIN*) 

• *ALL 

– Conditioned on replacement data (V5R4), Msg type and/or severity (7.1) 

 Watch for LIC Log entry (V5R4) on: 

– *ALL major and minor codes (but not *ALL *ALL) 

– Wildcard major and minor codes 

– Specific major and minor codes 

– Conditioned on replacement data 

 Watch for PAL entry (V6R1) on: 

– *ALL System Reference Codes (SRCs) 

– Wildcard and Generic SRCs 

– Conditioned on replacement data – Resource name, type, or model 

Message Watch Exit 

Program 

 Reason for call (Input) 
– *MSGID (V5R4) 

– *LICLOG (V5R4) 

– *PAL (V6R1) 

– *STRWCH (V6R1) 

– *ENDWCH (V6R1) 

 Session ID (Input) 

 Return code (Output) 
– Error 

 Problem data (Input): 
– MSGID 

– Message queue 

– Job 

– Sending program/procedure 

– Receiving program/procedure 

– Replacement data 

– etc 

Message Watching 

Asynchronous 

Notification 

 Problem detection/isolation outside of 
executing job 
– Runs concurrent to job experiencing problem 

(good and bad) 

– No need to poll watched job (good 
performance) 

– Job Interrupt API (discussed next) 

 

 Sample scenario 
– CPF2456 - Log version &1 in &2 closed and 

should be saved.  

– Automate saving of log to *SAVF 

Message Watch 

Getting Started 

 Start the Watch 
 

– STRWCH SSNID(HISTORY)  WCHPGM(SAVELOG)  

          WCHMSG(CPF2456) WCHMSGQ(*HSTLOG) 

           

 Work with Watches (WRKWCH *ALL) 

 
                                                  Work with Watches                                 

                                                              

 Type options, press Enter.                                                      

   1=Start   2=End                                                               

                                                                                 

 Opt    Type/Session                          Origin              User             Status      

                                                                                 

          STRWCH                                                                   

            HISTORY                            STRWCH        VINING        ACTIVE      

Exit Program 

Data Definitions 
dSaveLog          pr                  extpgm('SAVELOG')         

d Type                          10    const                     

d Session                       10    const                     

d Error                         10                              

d Data                                likeds(DataDS)            

                                                                

dSaveLog          pi                                            

d Type                          10    const                     

d Session                       10    const                     

d Error                         10                              

d Data                                likeds(DataDS)            

                                                                

dExcCmd           pr                  extpgm('QCMDEXC')         

d Cmd                          100    const                     

d CmdLen                        15  5 const                     

                                                                

dDataDS           ds          1000                            

d MsgID                   5     11                            

d OffRplDta             441    444i 0                         

                                                              

dMsgDtaPtr        s               *                           

dMsgDta           ds            75    based(MsgDtaPtr)        

d File                    1     10                            

d Lib                    11     20                            

                                                              

dCmd              s             80  

Copyright Bruce Vining Services LLC, 2010, 2012 

All  Rights Reserved 

Message Watch 

Exit Program 
 /free                                                                  

                                                                        

  if Type = '*MSGID';                       // Make sure message watch  

                                                                        

     if Data.MsgID = 'CPF2456';             // Make sure for CPF2456 

        MsgDtaPtr = %addr(Data) + Data.OffRplDta;                       

        Cmd = 'SAVOBJ OBJ(' + File + ') LIB(' + Lib +                   

              ') DEV(*SAVF) SAVF(QGPL/MYSAVES)';                        

        ExcCmd(Cmd :%len(%trimr(Cmd)));     // Save the log  

            

     else;                                  // Not CPF2456              

        dsply ('Wrong Msg:' + Data.MsgID);                              

     endif; 

                                                             

  else;                                     // Not a *MSGID watch       

     dsply ('Wrong Type:' + Type);                                      

  endif;                                                                

                                                                        

  Error = *blanks;                                                      

  return;               

  *inlr = *on; 

 

 /end-free 

Call Job Interrupt 

Program (QWCJBITP) 

 Allows you to run an exit program in the initial thread 
of another job 
– See what’s in the QTEMP library 
– Update objects such as a *DTAARA in QTEMP 
– Basically run whatever the current caller of the API is 

authorized to 
– Can be very dangerous when in the wrong hands…. 

 
 New system value QALWJOBITP 

– Never allow 
– Default new jobs to not allow 
– Default new jobs to allow 

 New system exit point QIBM_QWC_JOBITPPGM 
 New Change Job Interrupt Status (QWCCJITP) API 
 Sample scenario – run commands in another job 

Call Job Interrupt  

Sample Command 

             CMD        PROMPT('Call Job Interrupt Program')         

             PARM       KWD(JOB) TYPE(Q1) PROMPT('Job name')         

             PARM       KWD(CMD) TYPE(*CMDSTR) LEN(512) MIN(1) +     

                          PROMPT('Command')                          

 Q1:         QUAL       TYPE(*NAME) LEN(10) MIN(1)                   

             QUAL       TYPE(*NAME) LEN(10) MIN(1) PROMPT('User')    

             QUAL       TYPE(*CHAR) LEN(6) RANGE(000000 999999) +    

                          MIN(1) FULL(*YES) PROMPT('Number') 

  ------------------------------------------------------                   
 

                    Call Job Interrupt Program (CALLJOB)                        

                                                                                

 Type choices, press Enter.                                                     

                                                                                

 Job name . . . . . . . . . . . . __________      Name                          

   User . . . . . . . . . . . . .   __________    Name                          

   Number . . . . . . . . . . . .   ______        000000-999999                 

 Command  . . . . . . . . . . . . _______________________                                                                                                                        

Call Job Interrupt  

Sample CPP  

 
dJobInt           pr                  extpgm('JOBINT')       

d Job                           26    const                  

d Cmd                          512    const                  

                                                             

dJobInt           pi                                         

d Job                           26    const                  

d Cmd                          512    const                  

 

dInterruptJob     pr                  extpgm('QWCJBITP')          

d Controls                            likeds(Controls) const      

d CtlFormat                      8    const                       

d ErrorCode                     10i 0 const                       

                                                                  

dControls         ds                                              

d PgmName                       10    inz('RUNCMD')               

d PgmLib                        10    inz('VINING')               

d TgtJob                        26                    

d Reserved                       2    inz(*ALLx'00')  

d OffPgmDta                     10i 0                 

d LenPgmDta                     10i 0                 

d PgmDta                       512 

Call Job Interrupt 

Sample CPP 
/free                                                                 

                                                                      

 TgtJob = Job;                             // Set target Job ID       

                                                                      

 LenPgmDta = %len(%trim(Cmd));             // Set length of command   

                                           //   to execute            

                                                                      

 PgmDta = %subst(Cmd :1 :LenPgmDta);       // Set command to execute  

 OffPgmDta = (%addr(PgmDta) - %addr(Controls));  // and offset to cmd 

                                                                      

 InterruptJob( Controls :'JITP0100' :0);   // Call the program in     

                                           //   the target job        

                                                                      

 *inlr = *on;                                                         

 return;                                                              

                                                                      

/end-free  

Code related to 

calling a command 

dRunCmd           pr                  extpgm('RUNCMD')         

d PgmDta                       512    const                    

d LenPgmDta                     10i 0 const                    

                                                               

dRunCmd           pi                                           

d PgmDta                       512    const                    

d LenPgmDta                     10i 0 const                    

 

dProcessCmd       pr                  extpgm('QCAPCMD')       

d PgmDta                       512    const                   

d LenPgmDta                     10i 0 const                   

d Options                             const likeds(Options)   

d LenOptions                    10i 0 const                   

d Format                         8    const                   

d NewCmd                         1                            

d LenAvlNewCmd                  10i 0 const                   

d LenNewCmd                     10i 0                       

d ErrorCode                     10i 0 const                 

                                                            

dOptions          ds                                        

d CmdType                       10i 0 inz(0)                

d DBCS                           1    inz('0')              

d Prompter                       1    inz('0')              

d Syntax                         1    inz('0')              

d MsgKey                         4    inz('    ')           

d CCSID                         10i 0 inz(0)                

d                                5    inz(*ALLx'00')  

Exit Program 

Main Function 

dNewCmd           s              1                                    

dLenNewCmd        s             10i 0                                 

                                                                      

 /free                                                                

                                                                      

  ProcessCmd( PgmDta :LenPgmDta :Options :%size(Options) :'CPOP0100'  

             :NewCmd :0 :LenNewCmd :0);                               

                                                                      

  *inlr = *on;                                                        

  return;                                                             

                                                                      

 /end-free  

Call Job Interrupt  

Sample Results 

 Target job 
– DSPDTAARA QTEMP/TEST 

–  Returns a value of ‘A’ 

 

 Source job 
– CALLJOB JOB(223306/VINING/QPADEV000C) 

CMD(CHGDTAARA DTAARA(QTEMP/TEST) 
VALUE('S')) 

 

 Target job 
– DSPDTAARA QTEMP/TEST 

– Returns a value of ‘S’  

Enhanced Integration 

at the User Interface 

 
 User Customized Options for WRKOUTQ, WRKSPLF, D P 

User Defined Options 

on Selected IBM Panels 

 Options A through Z can be defined 

– ‘A’ for Archive 

– ‘P’ for Priority or Print Next 

– Pick your own… 

 Exit programs registered using Registration Facility 

– ADDEXITPGM EXITPNT(QIBM_QSP_SPLF_LSTACT)  
FORMAT(LASP0100) PGMNBR(*LOW) 
PGM(VINING/SPLEXIT) PGMDTA(*JOB 1 P) 

• One exit program per exit point option (P in this example) 

• 26 exit programs per exit point (A to Z) 

 Function is actually within UIM 

– Parameters to pass are defined in UIM source 

– Only current user is QIBM_QSP_SPLF_LSTACT (Spool  
List Action) 

– Let IBM know if you believe this should be extended to  
other areas 

Parameters Passed  

IBM_QSP_SPLF_LSTACT 

 Exit Point Name – QIBM_QSP_SPLF_LSTACT 
 Exit Point Format – LASP0100 
 Action – for example option P (Priority and/or Print Next) 
 Spool Identifiers – the exit program parameter that will vary 

across exits 
 Length of spool identifiers 

 
 Spool Identifiers: 

– Job name 
– Job user 
– Job number 
– Spool file name 
– Spool file number 
– Job system name 
– Date and time created 
– Qualified output queue name 

SPLEXIT Program in RPG:  

Support Option ‘P’ for 

Prioritize Spooled Output 
 

dSplExit          pr                  ExtPgm('SPLEXIT')           

d ExitPointName                 20    const                       

d Format                         8    const                       

d Action                         1    const                       

d SpoolID                             likeds(espp0100)            

d SpoolIDLen                    10i 0 const                       

                                                                  

dSplExit          pi                                              

d ExitPointName                 20    const                       

d Format                         8    const                       

d Action                         1    const                       

d SpoolID                             likeds(espp0100)            

d SpoolIDLen                    10i 0 const                       

                                                                  

dMovSplf          pr                  ExtPgm('QSPMOVSP')          

d MoveInfo                   65535    const options(*varsize)     

d MoveInfoLen                   10i 0 const           

d MoveInfoFormat                 8    const           

d QUSEC                               likeds(qusec)  

 The Prototypes: 

Support Option ‘P’ for 

Prioritize Spooled Output  

dSndPgmMsg        pr                  extpgm('QMHSNDPM')           

d MsgID                          7    const                        

d QualMsgF                      20    const                        

d MsgDta                     65535    const options(*varsize)      

d LenMsgDta                     10i 0 const                        

d MsgType                       10    const                        

d CallStackEntry             65535    const options(*varsize)      

d CallStackCntr                 10i 0 const                        

d MsgKey                         4                                 

d QUSEC                               likeds(qusec)                

d LenCSE                        10i 0 const options(*nopass)       

d CSEQual                       20    const options(*nopass)       

d DSPWaitTime                   10i 0 const options(*nopass)       

d CSEType                       10    const options(*nopass)       

d CCSID                         10i 0 const options(*nopass)       

                                                                   

d/copy qsysinc/qrpglesrc,espxpts                                   

d/copy qsysinc/qrpglesrc,qspmovsp         

d/copy qsysinc/qrpglesrc,qusec            

                                          

dMsgText          s             80        

dMsgKey           s              4        

dWait             s              1  

 Prototype continued, QSYSINC Includes, Variables 

QSYSINC Definition 

Format LASP0100 

DESPP0100         DS                                                 

D*                                             Qsp LASP0100 t        

D ESPJN00                 1     10                                   

D*                                             Job Name              

D ESPUN00                11     20                                   

D*                                             User Name             

D ESPJNBR00              21     26                                   

D*                                             Job Number            

D ESPSN04                27     36                                   

D*                                             Splf Name              

D ESPSNBR02              37     40B 0                                

D*                                             Splf Number            

D ESPJSN03               41     48                                    

D*                                             Job System Name        

D ESPSCD03               49     55                                    

D*                                    Date in CYYMMDD format          

D ESPSCT03               56     61                                    

D*                                    Time in HHMMSS  format          

D ESPOQN02               62     71                                    

D*                                             Output Queue Name      

D ESPOQL02               72     81                                    

D*                                             Output Queue Lib  

QSYSINC Definition 

QSPMOVSP Format 

MSPF0100 
DQSPF0100         DS                                                

D*                                             Qsp MSPF0100         

D QSPSJN01                1     10                                  

D*                                             Source Job Name      

D QSPSJUN01              11     20                                  

D*                                             Source Job User Name  

D QSPSJNBR01             21     26                                  

D*                                             Source Job Number    

D QSPSIJID01             27     42                                  

D*                                             Source Internal Job ID 

D QSPISFID               43     58                                  

D*                                             Source Internal Spool File ID 

D QSPSSFN                59     68                                  

D*                                             Source Spooled File  Name 

D QSPSSFN00              69     72B 0                               

D*                                             Source Spooled File Number 

D QSPTOQN                73     82                                  

D*                                             Target Output Queue Name 

D QSPTOQL                83     92                                  

D*                                             Target Output Queue Library 

D QSPSJSN                93    100                                  

D*                                             Source Job System Name 

D QSPSSCD               101    107                                  

D*                                             Source Splf Create  Date  

D QSPERVED06            108    108                                  

D*                                             Reserved             

D QSPSSCT               109    114                                  

D*                                             Source Splf Create Time 

SPLEXIT Program 

The Code  
 

 /free                                                                   

  if ExitPointName = 'QIBM_QSP_SPLF_LSTACT';                             

     if Format = 'LASP0100';                                             

        if Action = 'P';                   // Priority request?         

           QSPSJN01   = SpoolID.ESPJN00;   // Job Name                   

           QSPSJUN01  = SpoolID.ESPUN00;   // User Name                  

           QSPSJNBR01 = SpoolID.ESPJNBR00; // Job Number                 

           QSPSIJID01 = *Blanks;           // Internal Job ID            

           QSPISFID   = *Blanks;           // Internal Spool ID          

           QSPSSFN    = SpoolID.ESPSN04;   // Spool File Name            

           QSPSSFN00  = SpoolID.ESPSNBR02; // Spool File Number          

           QSPTOQN    = '*SAME';           // Target Output Queue        

           QSPTOQL    = *Blanks;           // Target Output Queue Lib    

           QSPSJSN    = SpoolID.ESPJSN03;  // Job System Name            

           QSPSSCD    = SpoolID.ESPSCD03;  // Spool File Create Date     

           QSPSSCT    = SpoolID.ESPSCT03;  // Spool File Create Time     

           QUSBPRV = 0;                    // Error code Bytes Provided  

           MovSplF( QSPF0100 :%size(QSPF0100) :'MSPF0100' :QUSEC);    

           MsgText = 'Spool file ' + %trimr(QSPSSFN) +                

                     ' moved to top of output queue';                 

           SndPgmMsg( 'CPF9898' :'QCPFMSG   *LIBL'                    

                     :MsgText :%len(%trimr(MsgText))                  

                     :'*COMP' :'*PGMBDY' :1 :MsgKey :QUSEC);  

 Getting The Job Done 

More of the code 

           else;                                                           

           dsply ('Called with unexpected option:' + Action) ' ' Wait;  

        endif;                                                          

     else;                                                              

        dsply ('Called with unexpected format:' + Format) ' ' Wait;     

     endif;                                                             

  else;                                                                 

     dsply 'Unexpected caller' ' ' Wait;                                

  endif;                                                                

  *inlr = *on;                                                          

  return;                                                               

 /end-free  

  Error-Related Logic 

After the Exit Program Runs 

Completion Message Shown on Panel 

Set User Function: 

More Meaningful Operator Information   
  Just What is PGM – P18A54R? 

Set User Function: 

More Meaningful Operator Information (cont.) 
  Just What is PGM – P18A54R?  Change to say ‘Payroll’ 

Set User Function: 

More Meaningful Operator Information with  

Change Current Job API QWCCCJOB 

 

dChgCurJob        pr                  extpgm('QWCCCJOB')           

d ChgInf                     65535    const options(*varsize)      

d QUSEC                               likeds(QUSEC)                

                                                                   

d/copy qsysinc/qrpglesrc,qusec                                     

                                                                   

dChgInf           ds                                               

d NbrAtrs                       10i 0                              

d KeyAtr                        10i 0                              

d LenKeyDta                     10i 0                              

d KeyDta                        10                                 

                                                                   

dWait             s              1  

  Prototype and Data 

Set User Function: 

More Meaningful Operator Information with  

Change Current Job API 

 

/free                                                                   

 QUSBPRV = 0;                        // API to send escapes             

 NbrAtrs = 1;                        // Change one job value            

 KeyAtr = 3;                         // Change function name            

 LenKeyDta = 10;                     // Function name is 10 bytes       

 KeyDta = 'Payroll';                 // Current application is Payroll  

 ChgCurJob( ChgInf :QUSEC);          // Make the change                 

 dsply 'I am off running Payroll' ' ' Wait;                             

                                                                        

 *inlr = *on;                                                           

 return;                                                                

/end-free  

 Making the Change 

Set User Function:  

Change Current Job (QWCCCJOB) 

 Two New Keys Introduced 
– 3 – Set Function Name 

• Function type will be USR 

• Function name will be determined by caller 

– 4 – Reset Function Type and Name 
• Function type and name will be set to blanks 

 Function is Updated When Calling System 
Functions 

 Function is Retained When Calling 
Application Functions 

Message Handler: 

New Inquiry Message Exit Point 
 Eliminate operator involvement 
 Exit point QIBM_QMH_HDL_INQEXT 
 Works with *EXT job message queue 
 Runs after job INQMSGRPY processing 

– *DFT – use default message reply 
– *SYSRPYL – use system reply list 

 Intercepts normal display of Display Program Messages screen 
– Any number of exit programs can be registered 
– First exit program to respond to message ends calling of exit 

programs 
– If no exit program responds, system displays Display Program 

Messages screen 
– Send Reply API (QMHSNDRM) enhanced to support *EXT 

message queue 
 Exit program(s) registered to QIBM_QMH_REPLY_INQ continue to run 

– Reject response 
– Replace response 
– Etc. 

Inquiry Message 

      Exit Point – Conditional Responses 

 
dCatchINQ         pr                  ExtPgm('CATCHINQ')            

d CallType                      10i 0 const                         

d QualMsgQName                  20    const                         

d MsgKey                         4    const                         

d MsgID                          7    const                         

                                                                    

dCatchINQ         pi                                                

d CallType                      10i 0 const                         

d QualMsgQName                  20    const                         

d MsgKey                         4    const                         

d MsgID                          7    const                         

                                                                    

dSndRply          pr                  extpgm('QMHSNDRM')            

d MsgKey                         4    const                         

d QualMsgQName                  20    const                         

d ReplyText                  65535    const options(*varsize)       

d LenReplyText                  10i 0 const                    

d RmvMsg                        10    const                    

d QUSEC                               likeds(QUSEC)            

d CCSID                         10i 0 const options(*nopass)   

d AllowReject                   10    const options(*nopass)  

  Prototypes 

Inquiry Message 

Exit Point – Conditional Responses (cont.) 

d/copy qsysinc/qrpglesrc,qusec                                

                                                              

d                sds                                          

d CurUsr                358    367                            

                                                              

dWait             s              1                            

                                                              

dCancel           c                   const('C')              

dDump             c                   const('D')  

   Data Definitions 

Inquiry Message 

Exit Point – Conditional Responses (cont.)  

 

 /free                                                                   

  if CallType = 1;                    // Inquiry msg needing reply       

     if MsgID = 'RNQ0102';            // Attempt to divide by zero       

        QUSBPRV = 0;                  // API Error Code set to 0         

        Select;                                                          

          When CurUsr = 'JOE_USER';   // RPG dump for user               

               SndRply( MsgKey :QualMsgQName :Dump :1 :'*NO' :QUSEC);    

          When CurUsr = 'VINING';     // Cancel for Vining               

               SndRply( MsgKey :QualMsgQName :Cancel :1 :'*NO' :QUSEC);  

          Other;                      // Leave for next exit program          

          EndSl;                                                         

     endif;                           // Leave for next exit program       

  else;                                                                  

     dsply ('Unexpected call type of ' + %char(CallType)) ' ' Wait;      

  endif;                                                                 

                                                                         

  *inlr = *on;                                                           

  return;      

 /end-free  

  The Code 

Some of the new 

V5R4 Security APIs 

 Check Password Exit Point 
(QIBM_QSY_CHK_PASSWRD) 

 

– Allows for audit record to record non-
compliance of password 

– Does not allow password change to be stopped 

– Does get called for CRTUSRPRF, 
CHGUSRPRF, CHGPWD, QSYCHGPW 

 

 Cryptographic Key Management 

Some the new V5R4 

Service APIs 

 Control PTF Order (QESCPTFO)  

– Order Individual PTFs, Groups, Cum packages, PSP information 

• Electronic - *SAVF or *IMAGE 

• Media 

– Resume orders at later time 

– Cancel orders 

– Optional status updates with *DTAQ 

 Set User Policy (QPDETPOL) 

– Maximum PTF order size for electronic distribution for LAN and modem 

– Length of time to retain problem related information 

– Level of problem documentation to send 

 Retrieve Policy Data (QPDETRTV) 

 Collect Hung Job Service Documentation (QPDETHNG) 

– Collects information related to job that appears to be hung 

– Used by watch policy monitor when ENDJOBABN used 

 Change Contact Information (QEDCHGIN) 

 Retrieve Contact Information (QEDRTVCI) 

 Watches – End (QSCEWCH), Start (QSCSWCH), Exit Program (QPDETWCH) 

 Several other APIs related to collecting data related to software problems 

Some of the new 

V6R1 APIs 

 Change *DTAQ API (QMHQCDQ) 

– Automatic reclaim of storage when *DTAQ is empty 

– i5/OS locking of *DTAQ – recognize use of 
ALCOBJ/WRKOBJLCK by users 

– API only works on local *DTAQs 

 Exit Point for Change Journal Receiver 

– QIBM_QJO_CHG_JRNRCV 

– Exit program receives control when journal receiver is detached 
from journal 

 Open List of History Log (QMHOLHST) 

 Retrieve Printer File Attributes (QDFRPRTA) 

 Change Storage Pool API (QWCCHGPL) to manage storage pool 
size, activity level, and tuning values 

 Cryptographic Exit Points 

– Clear Master Key 

– Delete Key Store Record 

– Set Master Key 

– Translate Key Store 

More of the new 

V6R1 APIs 

 Verify Name API (QCAVFYNM ) 

 

 Delete Object API (QLIDLTO) 

 

 Change Program Data API (QBNCHGPD) 
– Update *Module, *Pgm, and *Srvpgm 

• Source file location (including debug view locations) 

• Bound *Module library names 

• Bound *Srvpgm library names 

– Prevent update of bound *SRVPGM library (6.1/7.1 PTF) 

 

 Create and Send MIME E-mail (QtmsCreateSendEmail) 
– Optionally sign and encrypt MIME E-mail using Secure MIME 

(S/MIME) 

– Attachment support from IFS stream file 
• What if? Watch for abnormal job termination, get joblog, send it! 

Some of the V6R1 

Changed Interfaces 

 
 Command Analyzer Retrieve Exit support for after 

command has run (5.4, 6.1, 7.1 PTF) 

 Retrieve Job Information (QUSRJOBI) and Open List of 
Jobs (QGYOLJOB) 

– Message key and qualified message queue of the 
message that an active job is waiting on 

– A real assist for automating of system operations 

 Send Reply Message (QMHSNDRM) 

– Support for *EXT for external message queue 

 Convert Edit Code (QECCVTEC) and Edit (QECEDT) 

– Support for up to 63 digits 

 Convert Date and Time (QWCCVTDT) 

– Year offsets 

Some of the new 

7.1 APIs 

 Spool File Security exit point 
– Function specific (Chg, Cpy, Dlt, Dsp, Hld, Mov, Rst, 

Save,Snd, etc) authorization control to a spool file 

 Retrieve Watch Information (QSCRWCHI) 

 Retrieve Watch List (QSCRWCHL) 

 Open List of Output Queues (QSPOLOTQ) 

 Retrieve Spool Information (QSPSPLI) 
– Number of spool files for a user or all users 

– Number of spool files on an ASP 

 Socket exit points to allow/disallow: 
– accept() 

– connect() 

– listen() 

Key Points to 

      Take Home 

 System APIs continue to grow in both 
callable interfaces and exit points 

 Most System APIs are created due to user 
requirements 

 Improvements target end user, 
development, and support 
– UIM customization exits 

– Message Handler exits 

– Job Interrupt 

– Message Watches 

Last Chance 

      Before the Break 

How to contact me: 

Bruce Vining 

bvining@brucevining.com 


